# $Id: PKGBUILD 160607 2012-06-02 21:00:23Z ibiru $
# Maintainer : Ionut Biru <ibiru@archlinux.org>
# Contributor: Marti Raudsepp <marti@juffo.org>

pkgname=media-player-info
pkgver=17
pkgrel=1
pkgdesc="Data files describing media player capabilities, for post-HAL systems"
arch=('any')
license=('BSD')
url="http://cgit.freedesktop.org/media-player-info/"
depends=('udev')
source=(http://www.freedesktop.org/software/media-player-info/$pkgname-$pkgver.tar.gz)
sha256sums=('ceab4699e27b683dec4f985be5667b684e1502728243a1bb711590a58ee31e46')

build() {
  cd $pkgname-$pkgver

  ./configure --prefix=/usr \
      --with-udevdir=/usr/lib/udev
  make
}

package() {
  cd "$pkgname-$pkgver"
  make DESTDIR="$pkgdir" install

  install -d "$pkgdir/usr/share/licenses/$pkgname"
  install -m644 COPYING "$pkgdir/usr/share/licenses/$pkgname"
}
